

## Logic programming

Lecture 23

Tuesday, April 13, 2021

### 1 Logic programming

Logic programming has its roots in automated theorem proving. Logic programming differs from theorem proving in that logic programming uses the framework of a logic to specify and perform computation. Essentially, a logic program computes values, using mechanisms that are also useful for deduction. Logic programming typically restricts itself to well-behaved fragments of logic.

We can think of logic programs as having two interpretations. In the *declarative interpretation*, a logic program declares *what* is being computed. In the *procedural interpretation*, a logic program describes *how* a computation takes place. In the declarative interpretation, one can reason about the correctness of a program without needing to think about underlying computation mechanisms; this makes declarative programs easier to understand, and to develop. A lot of the time, once we have developed a declarative program using a logic programming language, we also have an executable specification, that is, a procedural interpretation that tells us *how* to compute what we described. This is one of the appealing features of logic programming. (In practice, executable specifications obtained this way are often inefficient; an understanding of the underlying computational mechanism is needed to make the execution of the declarative program efficient.)

We'll consider some of the concepts of logic programming by considering the programming language Prolog, which was developed in the early 70s, initially as a programming language for natural language processing.

### 2 Prolog

We start by introducing some terminology and syntax.

Variables	$X, Y, Z \in \mathbf{Var}$
Predicate symbols	$p, q, r \in \mathbf{PSym}$
Function symbols	$f, g, h, a, b, c \in \mathbf{FSym}$
Terms	$s, t ::= X \mid f(t_1, \dots, t_n)$
Atoms	$A, H ::= p(t_1, \dots, t_n)$

We assume that each predicate symbol and function symbol has a fixed *arity*, that is, the number of arguments associated with it. A *constant* (ranged over by  $a, b, c$ ) is a function symbol with arity 0. For example, `alice` and `bob` are constants. Suppose predicate symbol `child` has arity 2; then `child(alice, bob)` is an atom, but `child(alice)` is not.

An *atom*  $p(t_1, \dots, t_n)$  is a predicate symbol  $p$  with an appropriate number of terms for the arity of the predicate symbol. Variables are meant to be place-holders for terms. We call a variable-free term a *ground term* and a variable-free atom a *ground atom*.

We don't assume any specific set of predicate symbols or function symbols. Indeed, the choice of symbols and their arity is part of the logic program. There may be some common symbols that are useful in many programs, such as a function symbol for integer addition. Function symbols do not have any implicit meaning, even function symbols for things like addition. We'll define the "meaning" of function symbols and predicate symbols shortly.

## 2.1 Unification

Variables are place-holders for terms. A *substitution* is a finite map from variables to terms. Much like the type substitutions we saw in Lecture 14, we can define the application of a substitution  $\sigma$  to terms and atoms, and define composition of substitutions.

Importantly, the concept of *unification* also applies to term substitutions. To recap: a substitution  $\sigma$  *unifies* terms  $s$  and  $t$  if  $\sigma(s) = \sigma(t)$ . Substitution  $\sigma$  is the *most general unifier* of terms  $s$  and  $t$  if for any substitution  $\sigma'$  that unifies  $s$  and  $t$ , we have  $\sigma' = \sigma \circ \sigma''$  for some  $\sigma''$ . That is, a substitution is the most general unifier of  $s$  and  $t$  if we can extend it to obtain any other unifier of  $s$  and  $t$ .

It may be the case that there is no unifier for some pairs of terms. For example, there is no substitution that unifies  $f(X, Y)$  and  $g(\text{alice})$ . In that case, we say that unification fails.

We can extend the notion of unification and most general unifiers to sets of term equations  $\{s_1 = t_1, \dots, s_n = t_n\}$ .

Substitutions are the result of logic program computations. We'll see this once we start defining logic programs, which we now address.

## 2.2 Clauses

A *clause* has the following form:

$$A_0 \text{ :- } A_1, \dots, A_m.$$

where  $A_0, A_1, \dots, A_m$  are atoms. We call  $A_0$  the *head* of the clause, and  $A_1, \dots, A_m$  the *body*. If the body is an empty sequence of atoms, then we write  $A_0$ .

Intuitively, we can think of a clause as meaning " $A_0$  is true if  $A_1, \dots, A_m$  are all true." We can think of a clause as being the logical formula

$$\forall X_1, \dots, X_k. A_1 \wedge \dots \wedge A_m \Rightarrow A_0$$

where  $X_1, \dots, X_k$  are the variables that appear in  $A_0$  and  $A_1, \dots, A_m$ .

A *logic program* is a set of clauses. A program is activated by providing an initial *query*, which is a sequence of atoms (often just a single atom). A solution to a query  $A$  is a substitution  $\sigma$  such that  $\sigma(A)$  is true.

For example, consider the following logic program, a set of simple clauses that define what it means for one list to be the concatenation of two other lists. We use  $[]$  as a 0-arity function symbol standing in for the empty list, and  $[t_1|t_2]$  as syntactic sugar for  $\text{cons}(t_1, t_2)$ , where function symbol  $\text{cons}$  is intended to be a list constructor:  $t_1$  is the head of the list, and  $t_2$  is the rest of the list. Similarly, we use  $[t_1, \dots, t_n]$  as sugar for the list of the  $n$  elements  $t_1, \dots, t_n$ .

```
append([], Y, Y).
append([H|T], Y, [H|U]) :- append(T, Y, U).
```

The query `append([alice], [bob], [alice, bob])` is true, since, based on our intuition of clauses, we have:

```
append([alice], [bob], [alice, bob]) is true if append([], [bob], [bob]) is true.
append([], [bob], [bob]) is true.
```

The result of the query `append([alice], [bob], [alice, bob])` is the empty substitution.

The query `append([alice], [bob], Z)` has the (unique) solution of the substitution  $[Z \mapsto [alice, bob]]$ .

The query `append([alice], Y, [bob])` has no solutions.

## 2.3 Example: type checking

Let's consider a more interesting example of a Prolog program. We will write a Prolog program that expresses the typing judgment for a simple lambda calculus. Interestingly, we can use the Prolog program to both check that a lambda term is well typed, and also to infer types for the lambda term!

First, we describe how we will encode a lambda calculus program as a Prolog term.

Here is the syntax of the lambda calculus and how we translate lambda calculus expressions to a Prolog term that our program will use. The function symbols `literal`, `plus`, `var`, `lambda`, and `apply` are used to encode the program.

$$\begin{aligned}
 e &::= n \mid x \mid \lambda x. e \mid e_1 e_2 \mid e_1 + e_2 \\
 \mathcal{T}[\![n]\!] &= \text{literal}(n) \\
 \mathcal{T}[\![x]\!] &= \text{var}(x) \\
 \mathcal{T}[\![\lambda x. e]\!] &= \text{lambda}(x, \mathcal{T}[\![e]\!]) \\
 \mathcal{T}[\![e_1 e_2]\!] &= \text{apply}(\mathcal{T}[\![e_1]\!], \mathcal{T}[\![e_2]\!]) \\
 \mathcal{T}[\![e_1 + e_2]\!] &= \text{plus}(\mathcal{T}[\![e_1]\!], \mathcal{T}[\![e_2]\!])
 \end{aligned}$$

We represent types using the constant `int` (for type **int**) and terms `arrow(s, t)` (for the type  $\tau_1 \rightarrow \tau_2$ , where  $\tau_1$  is represented by term  $s$  and  $\tau_2$  is represented by term  $t$ ).

We use predicate `type(G, E, T)` to mean that expression  $E$  has type  $T$  in variable context  $G$ . A context is a list of pairs that map variable names to types.

We can now define our Prolog<sup>1</sup> program that encodes, quite directly, the typing rules.

```

type(G, literal(X), int).
type(G, var(X), T) :- member([X, T], G).
type(G, apply(M, N), T) :- type(G, M, arrow(S, T)), type(G, N, S).
type(G, lambda(X, M), arrow(S, T)) :- type([[X, S]|G], M, T).

```

## 2.4 Computation

How do we go about finding a solution to a query? That is, what is the procedural interpretation of a logic program?

We start with the empty substitution. Given a query, we look to see what clauses have heads that can be unified with that query. If we find a clause with a head that has the same predicate symbol, then we try to unify the query with the head of the clause by extending our substitution; if they can be unified, then we regard each atom in the body of the clause as a query that we are trying to satisfy. We repeat, recursively, until either unification fails, or we have satisfied all queries, and thus have a solution (which is the substitution we have built up by successive unifications).

Now, for a given query, there may be many clauses that have a head with the same predicate symbol. How do we choose which clause to consider? In logic programming, the choice is nondeterministic: you just magically pick one that works.

However, for implementation purposes, we need some way to decide which clause to consider. In Prolog, the order that clauses are declared in is important: clauses are considered in the order that they are declared. Thus, in Prolog, a program is not a *set* of clauses, but rather a *list* of clauses. A depth-first strategy is used for evaluation in Prolog: when unification fails, we *backtrack* to the last choice point (a point in the computation at which there were more than one applicable clauses), and try a different choice; if there are no more choices, then we backtrack to previous choice point. If no choice point is left, then we fail. Backtracking to a choice point requires us to throw away the extensions to the substitution that we made after that choice point.

## 2.5 Termination

A Prolog program may fail to terminate. The termination of a recursive Prolog program depends on the order of the clauses, and on the order of atoms in clause bodies. For example, consider the following

<sup>1</sup>To avoid typing self application with a cyclic type, this program needs the flag `occurs_check` set to `true`.

program, that describes edges in a graph, and allows us to find which nodes are reachable from other nodes (i.e., the reflexive, transitive closure of the edge relation).

```
edge(alice, bob).
reach(X, X).
reach(X, Z) :- edge(X, Y), reach(Y, Z).
```

When executing the query `reach(alice, Y)`, Prolog finds 2 solutions,  $Y$  mapping to `alice` and to `bob`. But if we modify the order of the clauses, as shown below, Prolog fails to terminate (or, depending on the implementation, may halt with a fatal error). (Why is that? Try executing the logic program according to the rules for computation we defined above.)

```
edge(alice, bob).
reach(X, Z) :- reach(Y, Z), edge(X, Y).
reach(X, X).
```

Thus, the procedural interpretation of a Prolog program can be quite different from its logical interpretation: a Prolog programmer must be aware of *how* the program will execute in order to guarantee termination and efficiency.

### 3 Datalog

Datalog, like Prolog, is a logic programming language. However, the semantics of Datalog differ from the semantics of Prolog. Syntactically, Datalog is a subset of Prolog.

In a Datalog program, the order of clauses are not important: a Datalog program can be thought of as a set of clauses, rather than a list. Moreover, unlike Prolog queries, Datalog queries (on finite sets) are guaranteed to terminate. Indeed, evaluation of Datalog programs is polynomial in the size of the program. More specifically, it is possible to enumerate all the ground terms implied by a Datalog program (i.e., a set of clauses) in time polynomial in the number of clauses in the program. (The degree of the polynomial is essentially the maximum number of variables that appear in any rule.)

In order to achieve decidability, Datalog is less expressive than Prolog. In particular, Datalog makes the following restrictions.

- Datalog does not allow functions with arity greater than 0. That is, the syntax for terms permits only variables and constants.

$$\text{Terms} \quad s, t ::= X \mid f$$

So, for example, whereas `append(cons(alice, []), [], [])` is a syntactically valid Prolog atom, it is not syntactically valid Datalog.

- Datalog requires that any variable that appears in the head of a clause also appears in the body of the clause. For example, the clause `foo(X, Y) :- bar(X)` is not allowed in Datalog, since the variable  $Y$  appears in the head of the clause, but not the body.

(There are also some additional restrictions when we allow negation. That is, Datalog restricts how terms can be negated in order to ensure decidability.)